

```

;      DDT RELOCATOR PROGRAM, INCLUDED WITH THE MODULE TO PERFORM
;      THE MOVE FROM 200H TO THE DESTINATION ADDRESS
0016 =  VERSION EQU      22      ;2.2
;
;      COPYRIGHT (C) 1976, 1977, 1978, 1979, 1980
;      DIGITAL RESEARCH
;      BOX 579 PACIFIC GROVE
;      CALIFORNIA 93950
;
0100      ORG      100H
0200 =    STACK EQU      200H
0005 =    BDOS   EQU      0005H
0009 =    PRNT   EQU      9      ;BDOS PRINT FUNCTION
0200 =    MODULE EQU      200H   ;MODULE ADDRESS
;
0100 01      DB      01H      ;LXI INSTRUCTION
0101          DS      2      ;SPACE FOR ADDRESS
;
;      LXI      B,0      ;ADDRESS FIELD FILLED-IN WHEN MODULE BUILT
0103 C33D01      JMP      START
0106 434F505952      DB      'COPYRIGHT (C) 1980, DIGITAL RESEARCH
0130 4444542056SIGNON: DB      'DDT VERS '
0139 322E          DB      VERSION/10+'0','.'
013B 3224          DB      VERSION MOD 10 + '0','.'
013D 310002      START: LXI      SP,STACK
0140 C5          PUSH     B
0141 C5          PUSH     B
0142 113001      LXI      D,SIGNON
0145 0E09          MVI      C,PRNT
0147 CD0500      CALL     BDOS
014A C1          POP      B      ;RECOVER LENGTH OF MOVE
014B 210700      LXI      H,BDOS+2;ADDRESS FIELD OF JUMP TO BDOS (TOP MEMORY)
014E 7E          MOV      A,M      ;A HAS HIGH ORDER ADDRESS OF MEMORY TOP
014F 3D          DCR      A      ;PAGE DIRECTLY BELOW BDOS
0150 90          SUB      B      ;A HAS HIGH ORDER ADDRESS OF RELOC AREA
0151 57          MOV      D,A
0152 1E00          MVI      E,0      ;D,E ADDRESSES BASE OF RELOC AREA
0154 D5          PUSH     D      ;SAVE FOR RELOCATION BELOW
;
0155 210002      LXI      H,MODULE;READY FOR THE MOVE
0158 78      MOVE:  MOV      A,B      ;BC=0?
0159 B1          ORA      C
015A CA6501      JZ      RELOC
015D 0B          DCX      B      ;COUNT MODULE SIZE DOWN TO ZERO
015E 7E          MOV      A,M      ;GET NEXT ABSOLUTE LOCATION
015F 12          STAX     D      ;PLACE IT INTO THE RELOC AREA
0160 13          INX      D
0161 23          INX      H
0162 C35801      JMP      MOVE
;
RELOC:  ;STORAGE MOVED, READY FOR RELOCATION
;      HL ADDRESSES BEGINNING OF THE BIT MAP FOR RELOCATION
;
0165 D1          POP      D      ;RECALL BASE OF RELOCATION AREA
0166 C1          POP      B      ;RECALL MODULE LENGTH
0167 E5          PUSH     H      ;SAVE BIT MAP BASE IN STACK
0168 62          MOV      H,D      ;RELOCATION BIAS IS IN D
;

```

CP/M 2.2
 COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950
 SERIAL # L-756

```

0169 78      RELO:  MOV    A,B      ;BC=0?
016A B1      ORA     C
016B CA8701  JZ       ENDREL
;
; NOT END OF THE RELOCATION, MAY BE INTO NEXT BYTE OF BIT MAP
016E 0B      DCX     B      ;COUNT LENGTH DOWN
016F 7B      MOV     A,E
0170 E607    ANI     111B    ;0 CAUSES FETCH OF NEXT BYTE
0172 C27A01  JNZ     REL1
;
; FETCH BIT MAP FROM STACKED ADDRESS
0175 E3      XTHL
0176 7E      MOV     A,M      ;NEXT 8 BITS OF MAP
0177 23      INX     H
0178 E3      XTHL          ;BASE ADDRESS GOES BACK TO STACK
0179 6F      MOV     L,A      ;L HOLDS THE MAP AS WE PROCESS 8 LOCATIONS
017A 7D      REL1:  MOV     A,L
017B 17      RAL
017C 6F      MOV     L,A      ;CY SET TO 1 IF RELOCATION NECESSARY
017D D28301  JNC     REL2     ;BACK TO L FOR NEXT TIME AROUND
;
; CURRENT ADDRESS REQUIRES RELOCATION
0180 1A      LDAX    D
0181 84      ADD     H      ;APPLY BIAS IN H
0182 12      STAX    D
0183 13      REL2:  INX     D      ;TO NEXT ADDRESS
0184 C36901  JMP     RELO     ;FOR ANOTHER BYTE TO RELOCATE
;
ENDREL: ;END OF RELOCATION
0187 D1      POP     D      ;CLEAR STACKED ADDRESS
0188 2E00    MVI     L,0
018A E9      PCHL          ;GO TO RELOCATED PROGRAM
018B      END

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950
 SERIAL # _____

0005 BDOS
0169 RELO
0200 STACK

0187 ENDREL
017A REL1
013D START

0200 MODULE
0183 REL2
0016 VERSION

0158 MOVE
0165 RELOC

0009 PRNT
0130 SIGNON

COPYRIGHT © 1980
DIGITAL RESEARCH, INC.
P. O. BOX 579
PACIFIC GROVE, CA 93950
SERIAL # _____

